

NAG Fortran Library Routine Document

F01RGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F01RGF reduces the complex m by n ($m \leq n$) upper trapezoidal matrix A to upper triangular form by means of unitary transformations.

2 Specification

```
SUBROUTINE F01RGF(M, N, A, LDA, THETA, IFAIL)
INTEGER          M, N, LDA, IFAIL
complex        A(LDA, *), THETA(*)
```

3 Description

The m by n ($m \leq n$) upper trapezoidal matrix A given by

$$A = (U \ X),$$

where U is an m by m upper triangular matrix, is factorized as

$$A = (R \ 0)P^H,$$

where P is an n by n unitary matrix and R is an m by m upper triangular matrix.

P is given as a sequence of Householder transformation matrices

$$P = P_m \cdots P_2 P_1,$$

the $(m - k + 1)$ th transformation matrix, P_k , being used to introduce zeros into the k th row of A . P_k has the form

$$P_k = \begin{pmatrix} I & 0 \\ 0 & T_k \end{pmatrix},$$

where

$$T_k = I - \gamma_k u_k u_k^H,$$

$$u_k = \begin{pmatrix} \zeta_k \\ 0 \\ z_k \end{pmatrix},$$

γ_k is a scalar for which $\text{Re } \gamma_k = 1.0$, ζ_k is a real scalar and z_k is an $(n - m)$ element vector. γ_k , ζ_k and z_k are chosen to annihilate the elements of the k th row of X and to make the diagonal elements of R real.

The scalar γ_k and the vector u_k are returned in the k th element of the array THETA and in the k th row of A , such that θ_k , given by

$$\theta_k = (\zeta_k, \text{Im } \gamma_k),$$

is in THETA(k) and the elements of z_k are in A($k, m + 1$), ..., A(k, n). The elements of R are returned in the upper triangular part of A .

For further information on this factorization and its use see Section 6.5 of Golub and van Loan (1996).

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1965) *The Algebraic Eigenvalue Problem* Oxford University Press, Oxford

5 Parameters

1: M – INTEGER *Input*

On entry: m , the number of rows of A .

When $M=0$ then an immediate return is effected.

Constraint: $M \geq 0$.

2: N – INTEGER *Input*

On entry: n , the number of columns of A .

Constraint: $N \geq M$.

3: A(LDA,*) – **complex** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, N)$.

On entry: the leading m by n upper trapezoidal part of the array A must contain the matrix to be factorized.

On exit: the m by m upper triangular part of A will contain the upper triangular matrix R , and the m by $(n - m)$ upper trapezoidal part of A will contain details of the factorization as described in Section 3.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F01RGF is called.

Constraint: $LDA \geq \max(1, M)$.

5: THETA(*) – **complex** array *Output*

Note: the dimension of the array THETA must be at least $\max(1, M)$.

On exit: THETA(k) contains the scalar θ_k for the $(m - k + 1)$ th transformation. If $T_k = I$, then THETA(k) = 0.0; if

$$T_k = \begin{pmatrix} \alpha & 0 \\ 0 & I \end{pmatrix}, \quad \text{Re } \alpha < 0.0$$

then THETA(k) = α , otherwise THETA(k) contains θ_k as described in Section 3 and $\text{Re}(\theta_k)$ is always in the range $(1.0, \sqrt{2.0})$.

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by $X04AAF$).

Errors or warnings detected by the routine:

$IFAIL = -1$

On entry, $M < 0$,
or $N < M$,
or $LDA < M$.

7 Accuracy

The computed factors R and P satisfy the relation

$$(R \ 0)P^H = A + E,$$

where

$$\|E\| \leq c\epsilon\|A\|,$$

ϵ is the *machine precision* (see $X02AJF$), c is a modest function of m and n , and $\|\cdot\|$ denotes the spectral (two) norm.

8 Further Comments

The approximate number of floating-point operations is given by $8m^2(n - m)$.

9 Example

To reduce the 3 by 4 matrix

$$\begin{pmatrix} 2.4 & 0.8 + 0.8i & -1.4 + 0.6i & 3.0 - 1.0i \\ 0 & 1.6 & 0.8 + 0.3i & 0.4 + 0.5i \\ 0 & 0 & 1.0 & 2.0 - 1.0i \end{pmatrix}$$

to upper triangular form.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F01RGF Example Program Text
*      Mark 14 release.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, LDA
      PARAMETER       (MMAX=10,NMAX=20,LDA=MMAX)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, J, M, N
*      .. Local Arrays ..
      complex         A(LDA,NMAX), THETA(MMAX)
*      .. External Subroutines ..
      EXTERNAL        F01RGF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F01RGF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      WRITE (NOUT,*)
      IF ((M.GT.MMAX) .OR. (N.GT.NMAX)) THEN
```

```

      WRITE (NOUT,*) 'M or N is out of range.'
      WRITE (NOUT,99999) 'M = ', M, ' N = ', N
    ELSE
      READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
      IFAIL = 0
*
*      Find the RQ factorization of A
      CALL F01RGF(M,N,A,LDA,THETA,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'RQ factorization of A'
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Vector THETA'
      WRITE (NOUT,99998) (THETA(I),I=1,M)
      WRITE (NOUT,*)
      WRITE (NOUT,*)
      + 'Matrix A after factorization (R is in left-hand upper triangle)'
      DO 20 I = 1, M
        WRITE (NOUT,99998) (A(I,J),J=1,N)
      20 CONTINUE
      END IF
*
99999 FORMAT (1X,A,I5,A,I5)
99998 FORMAT (1X,4(' (',F7.4,',',F8.4,')',:))
      END

```

9.2 Program Data

F01RGF Example Program Data

```

  3      4                                     :Values of M and N
( 2.4, 0.0 ) ( 0.8, 0.8 ) (-1.4, 0.6 ) ( 3.0,-1.0 )
( 0 , 0 ) ( 1.6, 0.0 ) ( 0.8, 0.3 ) ( 0.4, 0.5 )
( 0 , 0 ) ( 0 , 0 ) ( 1.0, 0.0 ) ( 2.0,-1.0 ) :End of matrix A

```

9.3 Program Results

F01RGF Example Program Results

RQ factorization of A

Vector THETA

```
( 1.2924, -0.0000 ) ( 1.3861, -0.0000 ) ( 1.1867, -0.0000 )
```

Matrix A after factorization (R is in left-hand upper triangle)

```
(-3.5808, 0.0000) ( 0.2533, -0.9059) (-2.2862, -0.6532) ( 0.5120, 0.2601)
( 0.0000, 0.0000) (-1.7369, 0.0000) (-0.4491, -0.6940) (-0.2544, -0.1187)
( 0.0000, 0.0000) ( 0.0000, 0.0000) (-2.4495, 0.0000) ( 0.6880, 0.3440)
```
